Rules for integrands of the form $u\,(e+f\,x)^m\,(a+b\,\text{Hyper}[c+d\,x])^p$

1. $\int \frac{(e+f\,x)^m\,\text{Hyper}[c+d\,x]^n}{a+b\,\text{Sinh}[c+d\,x]}\,dx$

1: $\int \frac{(e+f\,x)^m\,\text{Sinh}[c+d\,x]^n}{a+b\,\text{Sinh}[c+d\,x]}\,dx$ when $(m\mid n)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{z^n}{a+b\,z}==\frac{z^{n-1}}{b}-\frac{a\,z^{n-1}}{b\,(a+b\,z)}$

–

Rule: If $(m\mid n)\in\mathbb{Z}^+$, then

$$\int \frac{(e+f\,x)^m\,\text{Sinh}[c+d\,x]^n}{a+b\,\text{Sinh}[c+d\,x]}\,dx \longrightarrow \frac{1}{b}\int (e+f\,x)^m\,\text{Sinh}[c+d\,x]^{n-1}\,dx-\frac{a}{b}\int \frac{(e+f\,x)^m\,\text{Sinh}[c+d\,x]^{n-1}}{a+b\,\text{Sinh}[c+d\,x]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sinh[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Sinh[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Cosh[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Cosh[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

2. $\int \frac{(e+f x)^m \operatorname{Cosh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $n \in \mathbb{Z}^+$

1. $\int \frac{(e+f x)^m \operatorname{Cosh}[c+d x]}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+$

1: $\int \frac{(e+f x)^m \operatorname{Cosh}[c+d x]}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+ \land a^2+b^2 == 0$

Derivation: Algebraic expansion

Basis: If $a^2+b^2 == 0$, then $\frac{\operatorname{Cosh}[z]}{a+b \operatorname{Sinh}[z]} == \frac{1}{b} - \frac{2}{b-a e^z} == -\frac{1}{b} + \frac{2 e^z}{a+b e^z}$

Basis: If $a^2-b^2 == 0$, then $\frac{\operatorname{Sinh}[z]}{a+b \operatorname{Cosh}[z]} == \frac{1}{b} - \frac{2}{b+a e^z} == -\frac{1}{b} + \frac{2 e^z}{a+b e^z}$

Note: Although the first expansion is simpler, the second is used so the antiderivative will be expressed in terms of $e^{c+d x}$ rather than $e^{-(c+d x)}$.

Rule: If $m \in \mathbb{Z}^+ \land a^2+b^2 == 0$, then

$$\int \frac{(e+f x)^m \operatorname{Cosh}[c+d x]}{a+b \operatorname{Sinh}[c+d x]} \, dx \longrightarrow -\frac{(e+f x)^{m+1}}{b f (m+1)} + 2 \int \frac{(e+f x)^m e^{c+d x}}{a+b e^{c+d x}} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]/(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  -(e+f*x)^(m+1)/(b*f*(m+1)) + 2*Int[(e+f*x)^m*E^(c+d*x)/(a+b*E^(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[a^2+b^2,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]/(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  -(e+f*x)^(m+1)/(b*f*(m+1)) + 2*Int[(e+f*x)^m*E^(c+d*x)/(a+b*E^(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(e+f\,x)^m \operatorname{Cosh}[c+d\,x]}{a+b \operatorname{Sinh}[c+d\,x]}\,dx$ when $m \in \mathbb{Z}^+ \wedge a^2+b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Cosh}[z]}{a+b\operatorname{Sinh}[z]} == \frac{1}{b} - \frac{1}{b-\left(a-\sqrt{a^2+b^2}\right)e^z} - \frac{1}{b-\left(a+\sqrt{a^2+b^2}\right)e^z} == -\frac{1}{b} + \frac{e^z}{a-\sqrt{a^2+b^2}+b\,e^z} + \frac{e^z}{a+\sqrt{a^2+b^2}+b\,e^z}$

Basis: $\frac{\operatorname{Sinh}[z]}{a+b\operatorname{Cosh}[z]} == \frac{1}{b} - \frac{1}{b+\left(a-\sqrt{a^2-b^2}\right)e^z} - \frac{1}{b+\left(a+\sqrt{a^2-b^2}\right)e^z} == -\frac{1}{b} + \frac{e^z}{a-\sqrt{a^2-b^2}+b\,e^z} + \frac{e^z}{a+\sqrt{a^2-b^2}+b\,e^z}$

Note: Although the first expansion is simpler, the second is used so the antiderivative will be expressed in terms of $e^{c+d\,x}$ rather than $e^{-(c+d\,x)}$.

Rule: If $m \in \mathbb{Z}^+ \wedge a^2+b^2 \neq 0$, then

$$\int \frac{(e+f\,x)^m \operatorname{Cosh}[c+d\,x]}{a+b\operatorname{Sinh}[c+d\,x]}\,dx \longrightarrow -\frac{(e+f\,x)^{m+1}}{b\,f\,(m+1)} + \int \frac{(e+f\,x)^m e^{c+d\,x}}{a-\sqrt{a^2+b^2}+b\,e^{c+d\,x}}\,dx + \int \frac{(e+f\,x)^m e^{c+d\,x}}{a+\sqrt{a^2+b^2}+b\,e^{c+d\,x}}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]/(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  -(e+f*x)^(m+1)/(b*f*(m+1)) +
  Int[(e+f*x)^m*E^(c+d*x)/(a-Rt[a^2+b^2,2]+b*E^(c+d*x)),x] +
  Int[(e+f*x)^m*E^(c+d*x)/(a+Rt[a^2+b^2,2]+b*E^(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[a^2+b^2,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]/(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  -(e+f*x)^(m+1)/(b*f*(m+1)) +
  Int[(e+f*x)^m*E^(c+d*x)/(a-Rt[a^2-b^2,2]+b*E^(c+d*x)),x] +
  Int[(e+f*x)^m*E^(c+d*x)/(a+Rt[a^2-b^2,2]+b*E^(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[a^2-b^2,0]
```

2. $\int \frac{(e+f x)^m \operatorname{Cosh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} dx$ when $n-1 \in \mathbb{Z}^+$

1: $\int \frac{(e+f x)^m \operatorname{Cosh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} dx$ when $n-1 \in \mathbb{Z}^+ \wedge a^2+b^2 == 0$

Derivation: Algebraic expansion

Basis: If $a^2+b^2 == 0$, then $\frac{\operatorname{Cosh}[z]^2}{a+b \operatorname{Sinh}[z]} == \frac{1}{a}+\frac{\operatorname{Sinh}[z]}{b}$

Basis: If $a^2-b^2 == 0$, then $\frac{\operatorname{Sinh}[z]^2}{a+b \operatorname{Cosh}[z]} == -\frac{1}{a}+\frac{\operatorname{Cosh}[z]}{b}$

Rule: If $n-1 \in \mathbb{Z}^+ \wedge a^2+b^2 == 0$, then

$$\int \frac{(e+f x)^m \operatorname{Cosh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} dx \longrightarrow \frac{1}{a} \int (e+f x)^m \operatorname{Cosh}[c+d x]^{n-2} dx+\frac{1}{b} \int (e+f x)^m \operatorname{Cosh}[c+d x]^{n-2} \operatorname{Sinh}[c+d x] dx$$

–

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]^n_/(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Cosh[c+d*x]^(n-2),x] +
  1/b*Int[(e+f*x)^m*Cosh[c+d*x]^(n-2)*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[n,1] && EqQ[a^2+b^2,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]^n_/(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  -1/a*Int[(e+f*x)^m*Sinh[c+d*x]^(n-2),x] +
  1/b*Int[(e+f*x)^m*Sinh[c+d*x]^(n-2)*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[n,1] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(e+f\,x)^m \operatorname{Cosh}[c+d\,x]^n}{a+b \operatorname{Sinh}[c+d\,x]} \, dx$ when $n-1 \in \mathbb{Z}^+ \wedge a^2+b^2 \neq 0 \wedge m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Cosh}[z]^2}{a+b \operatorname{Sinh}[z]} == -\frac{a}{b^2} + \frac{\operatorname{Sinh}[z]}{b} + \frac{a^2+b^2}{b^2\,(a+b \operatorname{Sinh}[z])}$

Basis: $\frac{\operatorname{Sinh}[z]^2}{a+b \operatorname{Cosh}[z]} == -\frac{a}{b^2} + \frac{\operatorname{Cosh}[z]}{b} + \frac{a^2-b^2}{b^2\,(a+b \operatorname{Cosh}[z])}$

Rule: If $n-1 \in \mathbb{Z}^+ \wedge a^2+b^2 \neq 0 \wedge m \in \mathbb{Z}^+$, then

$$\int \frac{(e+f\,x)^m \operatorname{Cosh}[c+d\,x]^n}{a+b \operatorname{Sinh}[c+d\,x]} \, dx \longrightarrow$$

$$-\frac{a}{b^2} \int (e+f\,x)^m \operatorname{Cosh}[c+d\,x]^{n-2} \, dx + \frac{1}{b} \int (e+f\,x)^m \operatorname{Cosh}[c+d\,x]^{n-2} \operatorname{Sinh}[c+d\,x] \, dx + \frac{a^2+b^2}{b^2} \int \frac{(e+f\,x)^m \operatorname{Cosh}[c+d\,x]^{n-2}}{a+b \operatorname{Sinh}[c+d\,x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]^n_/(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  -a/b^2*Int[(e+f*x)^m*Cosh[c+d*x]^(n-2),x] +
  1/b*Int[(e+f*x)^m*Cosh[c+d*x]^(n-2)*Sinh[c+d*x],x] +
  (a^2+b^2)/b^2*Int[(e+f*x)^m*Cosh[c+d*x]^(n-2)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[n,1] && NeQ[a^2+b^2,0] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]^n_/(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  -a/b^2*Int[(e+f*x)^m*Sinh[c+d*x]^(n-2),x] +
  1/b*Int[(e+f*x)^m*Sinh[c+d*x]^(n-2)*Cosh[c+d*x],x] +
  (a^2-b^2)/b^2*Int[(e+f*x)^m*Sinh[c+d*x]^(n-2)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[n,1] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

3: $\int \frac{(e+f x)^m \operatorname{Tanh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Tanh}[z]^p}{a+b \operatorname{Sinh}[z]} = \frac{\operatorname{Sech}[z] \operatorname{Tanh}[z]^{p-1}}{b} - \frac{a \operatorname{Sech}[z] \operatorname{Tanh}[z]^{p-1}}{b (a+b \operatorname{Sinh}[z])}$

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Tanh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx \rightarrow \frac{1}{b} \int (e+f x)^m \operatorname{Sech}[c+d x] \operatorname{Tanh}[c+d x]^{n-1} \, dx - \frac{a}{b} \int \frac{(e+f x)^m \operatorname{Sech}[c+d x] \operatorname{Tanh}[c+d x]^{n-1}}{a+b \operatorname{Sinh}[c+d x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Tanh[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sech[c+d*x]*Tanh[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Sech[c+d*x]*Tanh[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Coth[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Csch[c+d*x]*Coth[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Csch[c+d*x]*Coth[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

4: $\int \frac{(e+f x)^m \operatorname{Coth}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Coth}[z]^n}{a+b \operatorname{Sinh}[z]} == \frac{\operatorname{Coth}[z]^n}{a} - \frac{b \operatorname{Cosh}[z] \operatorname{Coth}[z]^{n-1}}{a (a+b \operatorname{Sinh}[z])}$

–

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Coth}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx \longrightarrow \frac{1}{a} \int (e+f x)^m \operatorname{Coth}[c+d x]^n \, dx - \frac{b}{a} \int \frac{(e+f x)^m \operatorname{Cosh}[c+d x] \operatorname{Coth}[c+d x]^{n-1}}{a+b \operatorname{Sinh}[c+d x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Coth[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Coth[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Cosh[c+d*x]*Coth[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Tanh[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Tanh[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Sinh[c+d*x]*Tanh[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

5. $\int \frac{(e + f x)^m \text{Sech}[c + d x]^n}{a + b \text{Sinh}[c + d x]} \, dx$ when $m \in \mathbb{Z}^+$

1: $\int \frac{(e + f x)^m \text{Sech}[c + d x]^n}{a + b \text{Sinh}[c + d x]} \, dx$ when $m \in \mathbb{Z}^+ \land a^2 + b^2 == 0$

Derivation: Algebraic expansion

Basis: If $a^2 + b^2 == 0$, then $\frac{1}{a+b \text{Sinh}[z]} == \frac{\text{Sech}[z]^2}{a} + \frac{\text{Sech}[z] \text{Tanh}[z]}{b}$

Basis: If $a^2 - b^2 == 0$, then $\frac{1}{a+b \text{Cosh}[z]} == -\frac{\text{Csch}[z]^2}{a} + \frac{\text{Csch}[z] \text{Coth}[z]}{b}$

Rule: If $m \in \mathbb{Z}^+ \land a^2 + b^2 == 0$, then

$$\int \frac{(e + f x)^m \text{Sech}[c + d x]^n}{a + b \text{Sinh}[c + d x]} \, dx \to \frac{1}{a} \int (e + f x)^m \text{Sech}[c + d x]^{n+2} \, dx + \frac{1}{b} \int (e + f x)^m \text{Sech}[c + d x]^{n+1} \text{Tanh}[c + d x] \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sech[c+d*x]^(n+2),x] +
  1/b*Int[(e+f*x)^m*Sech[c+d*x]^(n+1)*Tanh[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && EqQ[a^2+b^2,0]
```

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  -1/a*Int[(e+f*x)^m*Csch[c+d*x]^(n+2),x] +
  1/b*Int[(e+f*x)^m*Csch[c+d*x]^(n+1)*Coth[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(e+f x)^m \operatorname{Sech}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+ \land a^2+b^2 \neq 0 \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Sech}[z]^2}{a+b \operatorname{Sinh}[z]} == \frac{b^2}{(a^2+b^2)(a+b \operatorname{Sinh}[z])} + \frac{\operatorname{Sech}[z]^2 (a-b \operatorname{Sinh}[z])}{a^2+b^2}$

Basis: $\frac{\operatorname{Csch}[z]^2}{a+b \operatorname{Cosh}[z]} == \frac{b^2}{(a^2-b^2)(a+b \operatorname{Cosh}[z])} + \frac{\operatorname{Csch}[z]^2 (a-b \operatorname{Cosh}[z])}{a^2-b^2}$

Rule: If $m \in \mathbb{Z}^+ \land a^2+b^2 \neq 0 \land n \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Sech}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx \longrightarrow \frac{b^2}{a^2+b^2} \int \frac{(e+f x)^m \operatorname{Sech}[c+d x]^{n-2}}{a+b \operatorname{Sinh}[c+d x]} \, dx + \frac{1}{a^2+b^2} \int (e+f x)^m \operatorname{Sech}[c+d x]^n (a-b \operatorname{Sinh}[c+d x]) \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  b^2/(a^2+b^2)*Int[(e+f*x)^m*Sech[c+d*x]^(n-2)/(a+b*Sinh[c+d*x]),x] +
  1/(a^2+b^2)*Int[(e+f*x)^m*Sech[c+d*x]^n*(a-b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[a^2+b^2,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  b^2/(a^2-b^2)*Int[(e+f*x)^m*Csch[c+d*x]^(n-2)/(a+b*Cosh[c+d*x]),x] +
  1/(a^2-b^2)*Int[(e+f*x)^m*Csch[c+d*x]^n*(a-b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[a^2-b^2,0] && IGtQ[n,0]
```

6: $\int \frac{(e+f x)^m \operatorname{Csch}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Csch}[z]^n}{a+b \operatorname{Sinh}[z]} == \frac{\operatorname{Csch}[z]^n}{a} - \frac{b \operatorname{Csch}[z]^{n-1}}{a (a+b \operatorname{Sinh}[z])}$

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Csch}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx \longrightarrow \frac{1}{a} \int (e+f x)^m \operatorname{Csch}[c+d x]^n \, dx - \frac{b}{a} \int \frac{(e+f x)^m \operatorname{Csch}[c+d x]^{n-1}}{a+b \operatorname{Sinh}[c+d x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Csch[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Csch[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sech[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Sech[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

U: $\int \frac{(e+f\,x)^m\,\text{Hyper}[c+d\,x]^n}{a+b\,\text{Sinh}[c+d\,x]}\,dx$

Rule:

$$\int \frac{(e+f\,x)^m\,\text{Hyper}[c+d\,x]^n}{a+b\,\text{Sinh}[c+d\,x]}\,dx \longrightarrow \int \frac{(e+f\,x)^m\,\text{Hyper}[c+d\,x]^n}{a+b\,\text{Sinh}[c+d\,x]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*F[c+d*x]^n/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && HyperbolicQ[F]
```

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*F[c+d*x]^n/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && HyperbolicQ[F]
```

2. $\int \frac{(e+f x)^m \,\text{Hyper1}[c+d x]^n \,\text{Hyper2}[c+d x]^p}{a+b \,\text{Sinh}[c+d x]} \, dx$

1: $\int \frac{(e+f x)^m \,\text{Cosh}[c+d x]^p \,\text{Sinh}[c+d x]^n}{a+b \,\text{Sinh}[c+d x]} \, dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{z^n}{a+b z} == \frac{z^{n-1}}{b} - \frac{a z^{n-1}}{b (a+b z)}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \,\text{Cosh}[c+d x]^p \,\text{Sinh}[c+d x]^n}{a+b \,\text{Sinh}[c+d x]} \, dx \to \frac{1}{b} \int (e+f x)^m \,\text{Cosh}[c+d x]^p \,\text{Sinh}[c+d x]^{n-1} \, dx - \frac{a}{b} \int \frac{(e+f x)^m \,\text{Cosh}[c+d x]^p \,\text{Sinh}[c+d x]^{n-1}}{a+b \,\text{Sinh}[c+d x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]^p_.*Sinh[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Cosh[c+d*x]^p*Sinh[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Cosh[c+d*x]^p*Sinh[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]^p_.*Cosh[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sinh[c+d*x]^p*Cosh[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Sinh[c+d*x]^p*Cosh[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

2: $\int \frac{(e+f x)^m \operatorname{Sinh}[c+d x]^p \operatorname{Tanh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Tanh}[z]^p}{a+b \operatorname{Sinh}[z]} == \frac{\operatorname{Tanh}[z]^p}{b \operatorname{Sinh}[z]} - \frac{a \operatorname{Tanh}[z]^p}{b \operatorname{Sinh}[z] (a+b \operatorname{Sinh}[z])}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Sinh}[c+d x]^p \operatorname{Tanh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx \longrightarrow \frac{1}{b} \int (e+f x)^m \operatorname{Sinh}[c+d x]^{p-1} \operatorname{Tanh}[c+d x]^n \, dx - \frac{a}{b} \int \frac{(e+f x)^m \operatorname{Sinh}[c+d x]^{p-1} \operatorname{Tanh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]^p_.*Tanh[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sinh[c+d*x]^(p-1)*Tanh[c+d*x]^n,x] -
  a/b*Int[(e+f*x)^m*Sinh[c+d*x]^(p-1)*Tanh[c+d*x]^n/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]^p_.*Coth[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Cosh[c+d*x]^(p-1)*Coth[c+d*x]^n,x] -
  a/b*Int[(e+f*x)^m*Cosh[c+d*x]^(p-1)*Coth[c+d*x]^n/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

3: $\int \frac{(e+f x)^m \operatorname{Sech}[c+d x]^p \operatorname{Tanh}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Tanh}[z]^p}{a+b \operatorname{Sinh}[z]} == \frac{\operatorname{Sech}[z] \operatorname{Tanh}[z]^{p-1}}{b} - \frac{a \operatorname{Sech}[z] \operatorname{Tanh}[z]^{p-1}}{b (a+b \operatorname{Sinh}[z])}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+fx)^m \operatorname{Sech}[c+dx]^p \operatorname{Tanh}[c+dx]^n}{a+b\operatorname{Sinh}[c+dx]}\,dx \longrightarrow \frac{1}{b}\int (e+fx)^m \operatorname{Sech}[c+dx]^{p+1}\operatorname{Tanh}[c+dx]^{n-1}\,dx-\frac{a}{b}\int \frac{(e+fx)^m \operatorname{Sech}[c+dx]^{p+1}\operatorname{Tanh}[c+dx]^{n-1}}{a+b\operatorname{Sinh}[c+dx]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]^p_.*Tanh[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sech[c+d*x]^(p+1)*Tanh[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Sech[c+d*x]^(p+1)*Tanh[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]^p_.*Coth[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Csch[c+d*x]^(p+1)*Coth[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Csch[c+d*x]^(p+1)*Coth[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

4: $\int \frac{(e+fx)^m \operatorname{Cosh}[c+dx]^p \operatorname{Coth}[c+dx]^n}{a+b\operatorname{Sinh}[c+dx]}\,dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Coth}[z]^n}{a+b\operatorname{Sinh}[z]} == \frac{\operatorname{Coth}[z]^n}{a} - \frac{b\operatorname{Cosh}[z]\operatorname{Coth}[z]^{n-1}}{a\,(a+b\operatorname{Sinh}[z])}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+fx)^m \operatorname{Cosh}[c+dx]^p \operatorname{Coth}[c+dx]^n}{a+b\operatorname{Sinh}[c+dx]}\,dx \longrightarrow \frac{1}{a}\int (e+fx)^m \operatorname{Cosh}[c+dx]^p\operatorname{Coth}[c+dx]^n\,dx-\frac{b}{a}\int \frac{(e+fx)^m \operatorname{Cosh}[c+dx]^{p+1}\operatorname{Coth}[c+dx]^{n-1}}{a+b\operatorname{Sinh}[c+dx]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]^p_.*Coth[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Cosh[c+d*x]^p*Coth[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Cosh[c+d*x]^(p+1)*Coth[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]^p_.*Tanh[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sinh[c+d*x]^p*Tanh[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Sinh[c+d*x]^(p+1)*Tanh[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

5: $\int \frac{(e+f x)^m \operatorname{Csch}[c+d x]^p \operatorname{Coth}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Coth}[z]^n}{a+b \operatorname{Sinh}[z]} = \frac{\operatorname{Coth}[z]^n}{a} - \frac{b \operatorname{Coth}[z]^n}{a \operatorname{Csch}[z]\ (a+b \operatorname{Sinh}[z])}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Csch}[c+d x]^p \operatorname{Coth}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx \longrightarrow \frac{1}{a} \int (e+f x)^m \operatorname{Csch}[c+d x]^p \operatorname{Coth}[c+d x]^n \, dx - \frac{b}{a} \int \frac{(e+f x)^m \operatorname{Csch}[c+d x]^{p-1} \operatorname{Coth}[c+d x]^n}{a+b \operatorname{Sinh}[c+d x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]^p_.*Coth[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Csch[c+d*x]^p*Coth[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Csch[c+d*x]^(p-1)*Coth[c+d*x]^n/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]^p_.*Tanh[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sech[c+d*x]^p*Tanh[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Sech[c+d*x]^(p-1)*Tanh[c+d*x]^n/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

6: $\int \frac{(e+fx)^m \operatorname{Sech}[c+dx]^p \operatorname{Csch}[c+dx]^n}{a+b \operatorname{Sinh}[c+dx]} \, dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Csch}[z]^n}{a+b \operatorname{Sinh}[z]} == \frac{\operatorname{Csch}[z]^n}{a} - \frac{b \operatorname{Csch}[z]^{n-1}}{a (a+b \operatorname{Sinh}[z])}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+fx)^m \operatorname{Sech}[c+dx]^p \operatorname{Csch}[c+dx]^n}{a+b \operatorname{Sinh}[c+dx]} \, dx \longrightarrow \frac{1}{a} \int (e+fx)^m \operatorname{Sech}[c+dx]^p \operatorname{Csch}[c+dx]^n \, dx - \frac{b}{a} \int \frac{(e+fx)^m \operatorname{Sech}[c+dx]^p \operatorname{Csch}[c+dx]^{n-1}}{a+b \operatorname{Sinh}[c+dx]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]^p_.*Csch[c_.+d_.*x_]^n_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sech[c+d*x]^p*Csch[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Sech[c+d*x]^p*Csch[c+d*x]^(n-1)/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]^p_.*Sech[c_.+d_.*x_]^n_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Csch[c+d*x]^p*Sech[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Csch[c+d*x]^p*Sech[c+d*x]^(n-1)/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

U: $\int \frac{(e+f\,x)^m\,\text{Hyper1}[c+d\,x]^n\,\text{Hyper2}[c+d\,x]^p}{a+b\,\text{Sinh}[c+d\,x]}\,dx$

Rule:

$$\int \frac{(e+f\,x)^m\,\text{Hyper1}[c+d\,x]^n\,\text{Hyper2}[c+d\,x]^p}{a+b\,\text{Sinh}[c+d\,x]}\,dx \longrightarrow \int \frac{(e+f\,x)^m\,\text{Hyper1}[c+d\,x]^n\,\text{Hyper2}[c+d\,x]^p}{a+b\,\text{Sinh}[c+d\,x]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_.*G_[c_.+d_.*x_]^p_./(a_+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*F[c+d*x]^n*G[c+d*x]^p/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && HyperbolicQ[F] && HyperbolicQ[G]
```

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_.*G_[c_.+d_.*x_]^p_./(a_+b_.*Cosh[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*F[c+d*x]^n*G[c+d*x]^p/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && HyperbolicQ[F] && HyperbolicQ[G]
```

3: $\int \frac{(e+f x)^m \text{Hyper}[c+d x]^n}{a+b \text{Sech}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: $\frac{1}{a+b \text{Sech}[z]} == \frac{\text{Cosh}[z]}{b+a \text{Cosh}[z]}$

Rule: If $(m \mid n) \in \mathbb{Z}$, then

$$\int \frac{(e+f x)^m \text{Hyper}[c+d x]^n}{a+b \text{Sech}[c+d x]} \, dx \longrightarrow \int \frac{(e+f x)^m \text{Cosh}[c+d x] \text{Hyper}[c+d x]^n}{b+a \text{Cosh}[c+d x]} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Sech[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Cosh[c+d*x]*F[c+d*x]^n/(b+a*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && HyperbolicQ[F] && IntegersQ[m,n]
```

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Csch[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Sinh[c+d*x]*F[c+d*x]^n/(b+a*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && HyperbolicQ[F] && IntegersQ[m,n]
```

4: $\int \frac{(e+f x)^m \text{Hyper1}[c+d x]^n \text{Hyper2}[c+d x]^p}{a+b \text{Sech}[c+d x]} dx$ when $(m \mid n \mid p) \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: $\frac{1}{a+b \text{Sech}[z]} == \frac{\text{Cosh}[z]}{b+a \text{Cosh}[z]}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}$, then

$$\int \frac{(e+f x)^m \text{Hyper1}[c+d x]^n \text{Hyper2}[c+d x]^p}{a+b \text{Sech}[c+d x]} dx \longrightarrow \int \frac{(e+f x)^m \text{Cosh}[c+d x] \text{Hyper1}[c+d x]^n \text{Hyper2}[c+d x]^p}{b+a \text{Cosh}[c+d x]} dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_.*G_[c_.+d_.*x_]^p_./(a_+b_.*Sech[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Cosh[c+d*x]*F[c+d*x]^n*G[c+d*x]^p/(b+a*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && HyperbolicQ[F] && HyperbolicQ[G] && IntegersQ[m,n,p]
```

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_.*G_[c_.+d_.*x_]^p_./(a_+b_.*Csch[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Sinh[c+d*x]*F[c+d*x]^n*G[c+d*x]^p/(b+a*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && HyperbolicQ[F] && HyperbolicQ[G] && IntegersQ[m,n,p]
```

Rules for integrands involving hyperbolic functions

0. $\int \text{Sinh}[a+b\,x]^p\,\text{Hyper}[c+d\,x]^q\,dx$

1: $\int \text{Sinh}[a+b\,x]^p\,\text{Sinh}[c+d\,x]^q\,dx$ when $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[v]^p\,\text{Sinh}[w]^q == \frac{1}{2^{p+q}}\left(-e^{-v}+e^{v}\right)^p\left(-e^{-w}+e^{w}\right)^q$

Rule: If $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$, then

$$\int \text{Sinh}[a+b\,x]^p\,\text{Sinh}[c+d\,x]^q\,dx \longrightarrow \frac{1}{2^{p+q}}\int\left(-e^{-c-d\,x}+e^{c+d\,x}\right)^q\,\text{ExpandIntegrand}\left[\left(-e^{-a-b\,x}+e^{a+b\,x}\right)^p,\,x\right]dx$$

Program code:

```
Int[Sinh[a_.+b_.*x_]^p_.*Sinh[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(-E^(-c-d*x)+E^(c+d*x))^q,(-E^(-a-b*x)+E^(a+b*x))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

```
Int[Cosh[a_.+b_.*x_]^p_.*Cosh[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(E^(-c-d*x)+E^(c+d*x))^q,(E^(-a-b*x)+E^(a+b*x))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

2: $\int \text{Sinh}[a+b\,x]^p\,\text{Cosh}[c+d\,x]^q\,dx$ when $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[v]^p\,\text{Cosh}[w]^q == \frac{1}{2^{p+q}}\left(-e^{-v}+e^{v}\right)^p\left(e^{-w}+e^{w}\right)^q$

Rule: If $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$, then

$$\int \text{Sinh}[a+b\,x]^p\,\text{Cosh}[c+d\,x]^q\,dx \longrightarrow \frac{1}{2^{p+q}}\int\left(e^{-c-d\,x}+e^{c+d\,x}\right)^q\,\text{ExpandIntegrand}\left[\left(-e^{-a-b\,x}+e^{a+b\,x}\right)^p,\,x\right]dx$$

Program code:

```
Int[Sinh[a_.+b_.*x_]^p_.*Cosh[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(E^(-c-d*x)+E^(c+d*x))^q,(-E^(-a-b*x)+E^(a+b*x))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

```
Int[Cosh[a_.+b_.*x_]^p_.*Sinh[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(-E^(-c-d*x)+E^(c+d*x))^q,(E^(-a-b*x)+E^(a+b*x))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

3: $\int \text{Sinh}[a+b\,x]\,\text{Tanh}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[v]\,\text{Tanh}[w]=-\frac{e^{-v}}{2}+\frac{e^{v}}{2}+\frac{e^{-v}}{1+e^{2w}}-\frac{e^{v}}{1+e^{2w}}$

Basis: $\text{Cosh}[v]\,\text{Coth}[w]=\frac{e^{-v}}{2}+\frac{e^{v}}{2}-\frac{e^{-v}}{1-e^{2w}}-\frac{e^{v}}{1-e^{2w}}$

Rule: If $b^2-d^2\neq 0$, then

$$\int \text{Sinh}[a+b\,x]\,\text{Tanh}[c+d\,x]\,dx\to\int\left(-\frac{e^{-a-b\,x}}{2}+\frac{e^{a+b\,x}}{2}+\frac{e^{-a-b\,x}}{1+e^{2\,(c+d\,x)}}-\frac{e^{a+b\,x}}{1+e^{2\,(c+d\,x)}}\right)dx$$

Program code:

```
Int[Sinh[a_.+b_.*x_]*Tanh[c_.+d_.*x_],x_Symbol] :=
  Int[-E^(-(a+b*x))/2 + E^(a+b*x)/2 + E^(-(a+b*x))/(1+E^(2*(c+d*x))) - E^(a+b*x)/(1+E^(2*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

```
Int[Cosh[a_.+b_.*x_]*Coth[c_.+d_.*x_],x_Symbol] :=
  Int[E^(-(a+b*x))/2 + E^(a+b*x)/2 - E^(-(a+b*x))/(1-E^(2*(c+d*x))) - E^(a+b*x)/(1-E^(2*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

4: $\int \text{Sinh}[a+b\,x]\,\text{Coth}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[v]\,\text{Coth}[w]=-\frac{e^{-v}}{2}+\frac{e^{v}}{2}+\frac{e^{-v}}{1-e^{2w}}-\frac{e^{v}}{1-e^{2w}}$

Basis: $\text{Cosh}[v]\,\text{Tanh}[w]=\frac{e^{-v}}{2}+\frac{e^{v}}{2}-\frac{e^{-v}}{1+e^{2w}}-\frac{e^{v}}{1+e^{2w}}$

Rule: If $b^2-d^2\neq 0$, then

$$\int \text{Sinh}[a+b\,x]\,\text{Coth}[c+d\,x]\,dx \longrightarrow \int \left(-\frac{e^{-a-b\,x}}{2}+\frac{e^{a+b\,x}}{2}+\frac{e^{-a-b\,x}}{1-e^{2\,(c+d\,x)}}-\frac{e^{a+b\,x}}{1-e^{2\,(c+d\,x)}}\right) dx$$

–

Program code:

```
Int[Sinh[a_.+b_.*x_]*Coth[c_.+d_.*x_],x_Symbol] :=
  Int[-E^(-(a+b*x))/2 + E^(a+b*x)/2 + E^(-(a+b*x))/(1-E^(2*(c+d*x))) - E^(a+b*x)/(1-E^(2*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

```
Int[Cosh[a_.+b_.*x_]*Tanh[c_.+d_.*x_],x_Symbol] :=
  Int[E^(-(a+b*x))/2 + E^(a+b*x)/2 - E^(-(a+b*x))/(1+E^(2*(c+d*x))) - E^(a+b*x)/(1+E^(2*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

1: $\int \text{Sinh}\left[\frac{a}{c+d\,x}\right]^n dx$ when $n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F\left[\frac{a}{c+d\,x}\right] == -\frac{1}{d}\,\text{Subst}\left[\frac{F[a\,x]}{x^2},\,x,\,\frac{1}{c+d\,x}\right]\,\partial_x \frac{1}{c+d\,x}$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \text{Sinh}\left[\frac{a}{c+d\,x}\right]^n dx \longrightarrow -\frac{1}{d}\,\text{Subst}\left[\int \frac{\text{Sinh}[a\,x]^n}{x^2}\,dx,\,x,\,\frac{1}{c+d\,x}\right]$$

–

Program code:

```
Int[Sinh[a_./(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Sinh[a*x]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,c,d},x] && IGtQ[n,0]
```

```
Int[Cosh[a_./(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Cosh[a*x]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,c,d},x] && IGtQ[n,0]
```

2. $\int \text{Sinh}\left[\frac{a+b x}{c+d x}\right]^n dx$ when $n \in \mathbb{Z}^+$

1: $\int \text{Sinh}\left[\frac{a+b x}{c+d x}\right]^n dx$ when $n \in \mathbb{Z}^+ \land b c - a d \neq 0$

Derivation: Integration by substitution

Basis: $F\left[\frac{a+b x}{c+d x}\right] == -\frac{1}{d} \text{Subst}\left[\frac{F\left[\frac{b}{d}-\frac{(b c-a d) x}{d}\right]}{x^2}, x, \frac{1}{c+d x}\right] \partial_x \frac{1}{c+d x}$

Rule: If $n \in \mathbb{Z}^+ \land b c - a d \neq 0$, then

$$\int \text{Sinh}\left[\frac{a+b x}{c+d x}\right]^n dx \to -\frac{1}{d} \text{Subst}\left[\int \frac{\text{Sinh}\left[\frac{b}{d}-\frac{(b c-a d) x}{d}\right]^n}{x^2} dx, x, \frac{1}{c+d x}\right]$$

Program code:

```
Int[Sinh[e_.*(a_.+b_.*x_)/(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Sinh[b*e/d-e*(b*c-a*d)*x/d]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,0] && NeQ[b*c-a*d,0]
```

```
Int[Cosh[e_.*(a_.+b_.*x_)/(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Cosh[b*e/d-e*(b*c-a*d)*x/d]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,0] && NeQ[b*c-a*d,0]
```

2: $\int \text{Sinh}[u]^n \, dx$ when $n \in \mathbb{Z}^+ \land u == \frac{a+b x}{c+d x}$

Derivation: Algebraic normalization

Rule: If $n \in \mathbb{Z}^+ \land u == \frac{a+b x}{c+d x}$, then

$$\int \text{Sinh}[u]^n \, dx \longrightarrow \int \text{Sinh}\left[\frac{a+b x}{c+d x}\right]^n dx$$

Program code:

```
Int[Sinh[u_]^n_.,x_Symbol] :=
  With[{lst=QuotientOfLinearsParts[u,x]},
  Int[Sinh[(lst[[1]]+lst[[2]]*x)/(lst[[3]]+lst[[4]]*x)]^n,x]] /;
IGtQ[n,0] && QuotientOfLinearsQ[u,x]
```

```
Int[Cosh[u_]^n_.,x_Symbol] :=
  With[{lst=QuotientOfLinearsParts[u,x]},
  Int[Cosh[(lst[[1]]+lst[[2]]*x)/(lst[[3]]+lst[[4]]*x)]^n,x]] /;
IGtQ[n,0] && QuotientOfLinearsQ[u,x]
```

3. $\int u\, \text{Sinh}[v]^p\, \text{Hyper}[w]^q\, dx$

1. $\int u\, \text{Sinh}[v]^p\, \text{Sinh}[w]^q\, dx$

1: $\int u\, \text{Sinh}[v]^p\, \text{Sinh}[w]^q\, dx$ when $w == v$

Derivation: Algebraic simplification

Rule: If $w == v$, then

$$\int u\, \text{Sinh}[v]^p\, \text{Sinh}[w]^q\, dx \longrightarrow \int u\, \text{Sinh}[v]^{p+q}\, dx$$

Program code:

```
Int[u_.*Sinh[v_]^p_.*Sinh[w_]^q_.,x_Symbol] :=
  Int[u*Sinh[v]^(p+q),x] /;
EqQ[w,v]
```

```
Int[u_.*Cosh[v_]^p_.*Cosh[w_]^q_.,x_Symbol] :=
  Int[u*Cosh[v]^(p+q),x] /;
EqQ[w,v]
```

2: $\int \text{Sinh}[v]^p \, \text{Sinh}[w]^q \, dx$ when $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$, then

$$\int \text{Sinh}[v]^p \, \text{Sinh}[w]^q \, dx \longrightarrow \int \text{TrigReduce}\left[\text{Sinh}[v]^p \, \text{Sinh}[w]^q\right] dx$$

Program code:

```
Int[Sinh[v_]^p_.*Sinh[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[Sinh[v]^p*Sinh[w]^q,x],x] /;
IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

```
Int[Cosh[v_]^p_.*Cosh[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[Cosh[v]^p*Cosh[w]^q,x],x] /;
IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

3: $\int x^m \, \text{Sinh}[v]^p \, \text{Sinh}[w]^q \, dx$ when $m \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$, then

$$\int x^m \, \text{Sinh}[v]^p \, \text{Sinh}[w]^q \, dx \longrightarrow \int x^m \, \text{TrigReduce}\left[\text{Sinh}[v]^p \, \text{Sinh}[w]^q\right] dx$$

Program code:

```
Int[x_^m_.*Sinh[v_]^p_.*Sinh[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^m,Sinh[v]^p*Sinh[w]^q,x],x] /;
IGtQ[m,0] && IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

```
Int[x_^m_.*Cosh[v_]^p_.*Cosh[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^m,Cosh[v]^p*Cosh[w]^q,x],x] /;
IGtQ[m,0] && IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

2. $\int u \operatorname{Sinh}[v]^p \operatorname{Cosh}[w]^q \, dx$

1: $\int u \operatorname{Sinh}[v]^p \operatorname{Cosh}[w]^p \, dx$ when $w == v \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $\operatorname{Sinh}[z] \operatorname{Cosh}[z] == \frac{1}{2} \operatorname{Sinh}[2 z]$

Rule: If $w == v \land p \in \mathbb{Z}$, then

$$\int u \operatorname{Sinh}[v]^p \operatorname{Cosh}[w]^p \, dx \longrightarrow \frac{1}{2^p} \int u \operatorname{Sinh}[2 v]^p \, dx$$

Program code:

```
Int[u_.*Sinh[v_]^p_.*Cosh[w_]^p_.,x_Symbol] :=
  1/2^p*Int[u*Sinh[2*v]^p,x] /;
EqQ[w,v] && IntegerQ[p]
```

2: $\int \text{Sinh}[v]^p \, \text{Cosh}[w]^q \, dx$ when $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$, then

$$\int \text{Sinh}[v]^p \, \text{Cosh}[w]^q \, dx \longrightarrow \int \text{TrigReduce}\left[\text{Sinh}[v]^p \, \text{Cosh}[w]^q\right] dx$$

Program code:

```
Int[Sinh[v_]^p_.*Cosh[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[Sinh[v]^p*Cosh[w]^q,x],x] /;
IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

3: $\int x^m \, \text{Sinh}[v]^p \, \text{Cosh}[w]^q \, dx$ when $m \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$, then

$$\int x^m \, \text{Sinh}[v]^p \, \text{Cosh}[w]^q \, dx \longrightarrow \int x^m \, \text{TrigReduce}\left[\text{Sinh}[v]^p \, \text{Cosh}[w]^q\right] dx$$

Program code:

```
Int[x_^m_.*Sinh[v_]^p_.*Cosh[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^m,Sinh[v]^p*Cosh[w]^q,x],x] /;
IGtQ[m,0] && IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

3. $\int u \operatorname{Sinh}[v]^p \operatorname{Tanh}[w]^q \, dx$

1: $\int \operatorname{Sinh}[v] \operatorname{Tanh}[w]^n \, dx$ when $n > 0 \land w \neq v \land x \notin v - w$

Derivation: Algebraic expansion

Basis: $\operatorname{Sinh}[v] \operatorname{Tanh}[w] == \operatorname{Cosh}[v] - \operatorname{Cosh}[v - w] \operatorname{Sech}[w]$

Basis: $\operatorname{Cosh}[v] \operatorname{Coth}[w] == \operatorname{Sinh}[v] + \operatorname{Cosh}[v - w] \operatorname{Csch}[w]$

Rule: If $n > 0 \land w \neq v \land x \notin v - w$, then

$$\int \operatorname{Sinh}[v] \operatorname{Tanh}[w]^n \, dx \longrightarrow \int \operatorname{Cosh}[v] \operatorname{Tanh}[w]^{n-1} \, dx - \operatorname{Cosh}[v - w] \int \operatorname{Sech}[w] \operatorname{Tanh}[w]^{n-1} \, dx$$

Program code:

```
Int[Sinh[v_]*Tanh[w_]^n_.,x_Symbol] :=
  Int[Cosh[v]*Tanh[w]^(n-1),x] - Cosh[v-w]*Int[Sech[w]*Tanh[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

```
Int[Cosh[v_]*Coth[w_]^n_.,x_Symbol] :=
  Int[Sinh[v]*Coth[w]^(n-1),x] + Cosh[v-w]*Int[Csch[w]*Coth[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

4. $\int u\,\text{Sinh}[v]^p\,\text{Coth}[w]^q\,dx$

1: $\int \text{Sinh}[v]\,\text{Coth}[w]^n\,dx$ when $n>0 \land w\neq v \land x\notin v-w$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[v]\,\text{Coth}[w] == \text{Cosh}[v] + \text{Sinh}[v-w]\,\text{Csch}[w]$

Basis: $\text{Cosh}[v]\,\text{Tanh}[w] == \text{Sinh}[v] - \text{Sinh}[v-w]\,\text{Sech}[w]$

Rule: If $n>0 \land w\neq v \land x\notin v-w$, then

$$\int \text{Sinh}[v]\,\text{Coth}[w]^n\,dx \longrightarrow \int \text{Cosh}[v]\,\text{Coth}[w]^{n-1}\,dx + \text{Sinh}[v-w]\int \text{Csch}[w]\,\text{Coth}[w]^{n-1}\,dx$$

Program code:

```
Int[Sinh[v_]*Coth[w_]^n_.,x_Symbol] :=
  Int[Cosh[v]*Coth[w]^(n-1),x] + Sinh[v-w]*Int[Csch[w]*Coth[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

```
Int[Cosh[v_]*Tanh[w_]^n_.,x_Symbol] :=
  Int[Sinh[v]*Tanh[w]^(n-1),x] - Sinh[v-w]*Int[Sech[w]*Tanh[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

5. $\int u \operatorname{Sinh}[v]^p \operatorname{Sech}[w]^q \, dx$

1: $\int \operatorname{Sinh}[v] \operatorname{Sech}[w]^n \, dx$ when $n > 0 \land w \neq v \land x \notin v - w$

Derivation: Algebraic expansion

Basis: $\operatorname{Sinh}[v] \operatorname{Sech}[w] == \operatorname{Cosh}[v - w] \operatorname{Tanh}[w] + \operatorname{Sinh}[v - w]$

Basis: $\operatorname{Cosh}[v] * \operatorname{Csch}[w] == \operatorname{Cosh}[v - w] * \operatorname{Coth}[w] + \operatorname{Sinh}[v - w]$

Rule: If $n > 0 \land w \neq v \land x \notin v - w$, then

$$\int \operatorname{Sinh}[v] \operatorname{Sech}[w]^n \, dx \longrightarrow \operatorname{Cosh}[v - w] \int \operatorname{Tanh}[w] \operatorname{Sech}[w]^{n-1} \, dx + \operatorname{Sinh}[v - w] \int \operatorname{Sech}[w]^{n-1} \, dx$$

Program code:

```
Int[Sinh[v_]*Sech[w_]^n_.,x_Symbol] :=
  Cosh[v-w]*Int[Tanh[w]*Sech[w]^(n-1),x] + Sinh[v-w]*Int[Sech[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

```
Int[Cosh[v_]*Csch[w_]^n_.,x_Symbol] :=
  Cosh[v-w]*Int[Coth[w]*Csch[w]^(n-1),x] + Sinh[v-w]*Int[Csch[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

6. $\int u \,\text{Sinh}[v]^p \,\text{Csch}[w]^q \,dx$

1: $\int \text{Sinh}[v] \,\text{Csch}[w]^n \,dx$ when $n > 0 \land w \neq v \land x \notin v - w$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[v] \,\text{Csch}[w] == \text{Sinh}[v-w] \,\text{Coth}[w] + \text{Cosh}[v-w]$

Basis: $\text{Cosh}[v] \,\text{Sech}[w] == \text{Sinh}[v-w] \,\text{Tanh}[w] + \text{Cosh}[v-w]$

Rule: If $n > 0 \land w \neq v \land x \notin v - w$, then

$$\int \text{Sinh}[v] \,\text{Csch}[w]^n \,dx \longrightarrow \text{Sinh}[v-w] \int \text{Coth}[w] \,\text{Csch}[w]^{n-1} \,dx + \text{Cosh}[v-w] \int \text{Csch}[w]^{n-1} \,dx$$

Program code:

```
Int[Sinh[v_]*Csch[w_]^n_.,x_Symbol] :=
  Sinh[v-w]*Int[Coth[w]*Csch[w]^(n-1),x] + Cosh[v-w]*Int[Csch[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

```
Int[Cosh[v_]*Sech[w_]^n_.,x_Symbol] :=
  Sinh[v-w]*Int[Tanh[w]*Sech[w]^(n-1),x] + Cosh[v-w]*Int[Sech[w]^(n-1),x] /;
GtQ[n,0] && NeQ[w,v] && FreeQ[v-w,x]
```

4: $\int (e+f x)^m (a+b \operatorname{Sinh}[c+d x] \operatorname{Cosh}[c+d x])^n \, dx$

Derivation: Algebraic simplification

Basis: $\operatorname{Sinh}[z] \operatorname{Cosh}[z] == \frac{1}{2} \operatorname{Sinh}[2 z]$

Rule:

$$\int (e+f x)^m (a+b \operatorname{Sinh}[c+d x] \operatorname{Cosh}[c+d x])^n \, dx \longrightarrow \int (e+f x)^m \left(a+\frac{1}{2} b \operatorname{Sinh}[2 c+2 d x]\right)^n dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_+b_.*Sinh[c_.+d_.*x_]*Cosh[c_.+d_.*x_])^n_.,x_Symbol] :=
  Int[(e+f*x)^m*(a+b*Sinh[2*c+2*d*x]/2)^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

5: $\int x^m \left(a + b\,\text{Sinh}[c + d\,x]^2\right)^n dx$ when $a - b \neq 0 \land m \in \mathbb{Z}^+ \land n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $\text{Sinh}[z]^2 == \frac{1}{2}\,(-1 + \text{Cosh}[2\,z])$

Basis: $\text{Cosh}[z]^2 == \frac{1}{2}\,(1 + \text{Cosh}[2\,z])$

Note: This rule should be replaced with rules that directly reduce the integrand rather than transforming it using hyperbolic power expansion!

Rule: If $a - b \neq 0 \land m \in \mathbb{Z}^+ \land n \in \mathbb{Z}^-$, then

$$\int x^m \left(a + b\,\text{Sinh}[c + d\,x]^2\right)^n dx \longrightarrow \frac{1}{2^n} \int x^m\,(2\,a - b + b\,\text{Cosh}[2\,c + 2\,d\,x])^n\,dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*Sinh[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  1/2^n*Int[x^m*(2*a-b+b*Cosh[2*c+2*d*x])^n,x] /;
FreeQ[{a,b,c,d},x] && NeQ[a-b,0] && IGtQ[m,0] && ILtQ[n,0] && (EqQ[n,-1] || EqQ[m,1] && EqQ[n,-2])
```

```
Int[x_^m_.*(a_+b_.*Cosh[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  1/2^n*Int[x^m*(2*a+b+b*Cosh[2*c+2*d*x])^n,x] /;
FreeQ[{a,b,c,d},x] && NeQ[a-b,0] && IGtQ[m,0] && ILtQ[n,0] && (EqQ[n,-1] || EqQ[m,1] && EqQ[n,-2])
```

6: $\int \frac{(f+g\,x)^m}{a+b\,\text{Cosh}[d+e\,x]^2+c\,\text{Sinh}[d+e\,x]^2}\,dx$ when $m\in\mathbb{Z}^+\land a+b\neq 0\land a+c\neq 0$

Derivation: Algebraic simplification

Basis: $a+b\,\text{Cosh}[z]^2+c\,\text{Sinh}[z]^2 == \frac{1}{2}\,(2\,a+b-c+(b+c)\,\text{Cosh}[2\,z])$

Rule: If $m\in\mathbb{Z}^+\land a+b\neq 0\land a+c\neq 0$, then

$$\int \frac{(f+g\,x)^m}{a+b\,\text{Cosh}[d+e\,x]^2+c\,\text{Sinh}[d+e\,x]^2}\,dx \longrightarrow 2\int \frac{(f+g\,x)^m}{2\,a+b-c+(b+c)\,\text{Cosh}[2\,d+2\,e\,x]}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_./(a_.+b_.*Cosh[d_.+e_.*x_]^2+c_.*Sinh[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(2*a+b-c+(b+c)*Cosh[2*d+2*e*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[m,0] && NeQ[a+b,0] && NeQ[a+c,0]
```

```
Int[(f_.+g_.*x_)^m_.*Sech[d_.+e_.*x_]^2/(b_+c_.*Tanh[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(b-c+(b+c)*Cosh[2*d+2*e*x]),x] /;
FreeQ[{b,c,d,e,f,g},x] && IGtQ[m,0]
```

```
Int[(f_.+g_.*x_)^m_.*Sech[d_.+e_.*x_]^2/(b_.+a_.*Sech[d_.+e_.*x_]^2+c_.*Tanh[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(2*a+b-c+(b+c)*Cosh[2*d+2*e*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[m,0] && NeQ[a+b,0] && NeQ[a+c,0]
```

```
Int[(f_.+g_.*x_)^m_.*Csch[d_.+e_.*x_]^2/(c_+b_.*Coth[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(b-c+(b+c)*Cosh[2*d+2*e*x]),x] /;
FreeQ[{b,c,d,e,f,g},x] && IGtQ[m,0]
```

```
Int[(f_.+g_.*x_)^m_.*Csch[d_.+e_.*x_]^2/(c_.+b_.*Coth[d_.+e_.*x_]^2+a_.*Csch[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(2*a+b-c+(b+c)*Cosh[2*d+2*e*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[m,0] && NeQ[a+b,0] && NeQ[a+c,0]
```

7: $\int \frac{(e+f x)\,(A+B\,\mathrm{Sinh}[c+d x])}{(a+b\,\mathrm{Sinh}[c+d x])^2}\,dx$ when $a A+b B==0$

Derivation: Integration by parts

Basis: If $a A+b B==0$, then $\frac{(A+B\,\mathrm{Sinh}[c+d x])}{(a+b\,\mathrm{Sinh}[c+d x])^2}==\partial_x \frac{B\,\mathrm{Cosh}[c+d x]}{a d\,(a+b\,\mathrm{Sinh}[c+d x])}$

Rule: If $a A+b B==0$, then

$$\int \frac{(e+f x)\,(A+B\,\mathrm{Sinh}[c+d x])}{(a+b\,\mathrm{Sinh}[c+d x])^2}\,dx \longrightarrow \frac{B\,(e+f x)\,\mathrm{Cosh}[c+d x]}{a d\,(a+b\,\mathrm{Sinh}[c+d x])}-\frac{B f}{a d}\int \frac{\mathrm{Cosh}[c+d x]}{a+b\,\mathrm{Sinh}[c+d x]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)*(A_+B_.*Sinh[c_.+d_.*x_])/(a_+b_.*Sinh[c_.+d_.*x_])^2,x_Symbol] :=
  B*(e+f*x)*Cosh[c+d*x]/(a*d*(a+b*Sinh[c+d*x])) -
  B*f/(a*d)*Int[Cosh[c+d*x]/(a+b*Sinh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && EqQ[a*A+b*B,0]
```

```
Int[(e_.+f_.*x_)*(A_+B_.*Cosh[c_.+d_.*x_])/(a_+b_.*Cosh[c_.+d_.*x_])^2,x_Symbol] :=
  B*(e+f*x)*Sinh[c+d*x]/(a*d*(a+b*Cosh[c+d*x])) -
  B*f/(a*d)*Int[Sinh[c+d*x]/(a+b*Cosh[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && EqQ[a*A-b*B,0]
```

8: $\int (e+f\,x)^m\,\text{Sinh}\left[a+b\,(c+d\,x)^n\right]^p\,dx$ when $m\in\mathbb{Z}^+\land p\in\mathbb{Q}$

Derivation: Integration by linear substitution

Rule: If $m\in\mathbb{Z}^+\land p\in\mathbb{Q}$, then

$$\int (e+f\,x)^m\,\text{Sinh}\left[a+b\,(c+d\,x)^n\right]^p\,dx\,\longrightarrow\,\frac{1}{d^{m+1}}\,\text{Subst}\left[\int (d\,e-c\,f+f\,x)^m\,\text{Sinh}\left[a+b\,x^n\right]^p\,dx,\,x,\,c+d\,x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sinh[a_.+b_.*(c_+d_.*x_)^n_]^p_.,x_Symbol] :=
  1/d^(m+1)*Subst[Int[(d*e-c*f+f*x)^m*Sinh[a+b*x^n]^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && RationalQ[p]
```

```
Int[(e_.+f_.*x_)^m_.*Cosh[a_.+b_.*(c_+d_.*x_)^n_]^p_.,x_Symbol] :=
  1/d^(m+1)*Subst[Int[(d*e-c*f+f*x)^m*Cosh[a+b*x^n]^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && RationalQ[p]
```

9: $\int \text{Sech}[v]^m\,(a+b\,\text{Tanh}[v])^n\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\wedge m+n=0$

Derivation: Algebraic simplification

Basis: $\frac{a+b\,\text{Tanh}[z]}{\text{Sech}[z]}=a\,\text{Cosh}[z]+b\,\text{Sinh}[z]$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\wedge m+n=0$, then

$$\int \text{Sech}[v]^m\,(a+b\,\text{Tanh}[v])^n\,dx\longrightarrow\int (a\,\text{Cosh}[v]+b\,\text{Sinh}[v])^n\,dx$$

Program code:

```
Int[Sech[v_]^m_.*(a_+b_.*Tanh[v_])^n_.,x_Symbol] :=
  Int[(a*Cosh[v]+b*Sinh[v])^n,x] /;
FreeQ[{a,b},x] && IntegerQ[(m-1)/2] && EqQ[m+n,0]
```

```
Int[Csch[v_]^m_.*(a_+b_.*Coth[v_])^n_.,x_Symbol] :=
  Int[(b*Cosh[v]+a*Sinh[v])^n,x] /;
FreeQ[{a,b},x] && IntegerQ[(m-1)/2] && EqQ[m+n,0]
```

10: $\int u\,\text{Sinh}[a+b\,x]^m\,\text{Sinh}[c+d\,x]^n\,dx$ when $m\in\mathbb{Z}^+\land n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m\in\mathbb{Z}^+\land n\in\mathbb{Z}^+$, then

$$\int u\,\text{Sinh}[a+b\,x]^m\,\text{Sinh}[c+d\,x]^n\,dx\,\longrightarrow\,\int u\,\text{TrigReduce}\left[\text{Sinh}[a+b\,x]^m\,\text{Sinh}[c+d\,x]^n\right]dx$$

Program code:

```
Int[u_.*Sinh[a_.+b_.*x_]^m_.*Sinh[c_.+d_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[u,Sinh[a+b*x]^m*Sinh[c+d*x]^n,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[u_.*Cosh[a_.+b_.*x_]^m_.*Cosh[c_.+d_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[u,Cosh[a+b*x]^m*Cosh[c+d*x]^n,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0] && IGtQ[n,0]
```

11: $\int \operatorname{Sech}[a+b\,x]\,\operatorname{Sech}[c+d\,x]\,dx$ when $b^2-d^2==0\land b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Basis: If $b^2-d^2==0\land b\,c-a\,d\neq 0$, then
$\operatorname{Sech}[a+b\,x]\,\operatorname{Sech}[c+d\,x]==-\operatorname{Csch}\left[\frac{b\,c-a\,d}{d}\right]\operatorname{Tanh}[a+b\,x]+\operatorname{Csch}\left[\frac{b\,c-a\,d}{b}\right]\operatorname{Tanh}[c+d\,x]$

Rule: If $b^2-d^2==0\land b\,c-a\,d\neq 0$, then

$$\int \operatorname{Sech}[a+b\,x]\,\operatorname{Sech}[c+d\,x]\,dx\longrightarrow -\operatorname{Csch}\left[\frac{b\,c-a\,d}{d}\right]\int \operatorname{Tanh}[a+b\,x]\,dx+\operatorname{Csch}\left[\frac{b\,c-a\,d}{b}\right]\int \operatorname{Tanh}[c+d\,x]\,dx$$

Program code:

```
Int[Sech[a_.+b_.*x_]*Sech[c_+d_.*x_],x_Symbol] :=
  -Csch[(b*c-a*d)/d]*Int[Tanh[a+b*x],x] + Csch[(b*c-a*d)/b]*Int[Tanh[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

```
Int[Csch[a_.+b_.*x_]*Csch[c_+d_.*x_],x_Symbol] :=
  Csch[(b*c-a*d)/b]*Int[Coth[a+b*x],x] - Csch[(b*c-a*d)/d]*Int[Coth[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

12: $\int \text{Tanh}[a+b x]\, \text{Tanh}[c+d x]\, dx$ when $b^2-d^2==0 \land b c-a d\neq 0$

Derivation: Algebraic expansion

Basis: If $b^2-d^2==0$, then $\text{Tanh}[a+b x]\, \text{Tanh}[c+d x]==\frac{b}{d}-\frac{b}{d}\text{Cosh}\left[\frac{b c-a d}{d}\right]\text{Sech}[a+b x]\, \text{Sech}[c+d x]$

Rule: If $b^2-d^2==0 \land b c-a d\neq 0$, then

$$\int \text{Tanh}[a+b x]\, \text{Tanh}[c+d x]\, dx \longrightarrow \frac{b x}{d}-\frac{b}{d}\text{Cosh}\left[\frac{b c-a d}{d}\right]\int \text{Sech}[a+b x]\, \text{Sech}[c+d x]\, dx$$

Program code:

```
Int[Tanh[a_.+b_.*x_]*Tanh[c_+d_.*x_],x_Symbol] :=
  b*x/d - b/d*Cosh[(b*c-a*d)/d]*Int[Sech[a+b*x]*Sech[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

```
Int[Coth[a_.+b_.*x_]*Coth[c_+d_.*x_],x_Symbol] :=
  b*x/d + Cosh[(b*c-a*d)/d]*Int[Csch[a+b*x]*Csch[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

13: $\int u\,(a\,\text{Cosh}[v]+b\,\text{Sinh}[v])^n\,dx$ when $a^2-b^2==0$

Derivation: Algebraic simplification

Basis: If $a^2-b^2==0$, then $a\,\text{Cosh}[z]+b\,\text{Sinh}[z]==a\,e^{\frac{a z}{b}}$

Rule: If $a^2-b^2==0$, then

$$\int u\,(a\,\text{Cosh}[v]+b\,\text{Sinh}[v])^n\,dx \longrightarrow \int u\left(a\,e^{\frac{a v}{b}}\right)^n dx$$

Program code:

```
Int[u_.*(a_.*Cosh[v_]+b_.*Sinh[v_])^n_.,x_Symbol] :=
  Int[u*(a*E^(a/b*v))^n,x] /;
FreeQ[{a,b,n},x] && EqQ[a^2-b^2,0]
```

14. $\int u\,\mathrm{Sin}\left[d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2\right]dx$

1: $\int \mathrm{Sinh}\left[d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2\right]dx$

Derivation: Algebraic expansion

Basis: $\mathrm{Sinh}[z] == -\frac{e^{-z}}{2}+\frac{e^{z}}{2}$

Rule:

$$\int \mathrm{Sinh}\left[d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2\right]dx \longrightarrow \frac{1}{2}\int e^{-d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2}dx+\frac{1}{2}\int e^{d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2}dx$$

–

Program code:

```
Int[Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  -1/2*Int[E^(-d*(a+b*Log[c*x^n])^2),x] + 1/2*Int[E^(d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  1/2*Int[E^(-d*(a+b*Log[c*x^n])^2),x] + 1/2*Int[E^(d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int (e\,x)^m \operatorname{Sinh}\left[d\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)^2\right] dx$

Derivation: Algebraic expansion

Basis: $\operatorname{Sinh}[z] == -\frac{e^{-z}}{2}+\frac{e^{z}}{2}$

Rule:

$$\int (e\,x)^m \operatorname{Sinh}\left[d\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)^2\right] dx \longrightarrow \frac{1}{2}\int (e\,x)^m\, e^{-d\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)^2} dx+\frac{1}{2}\int (e\,x)^m\, e^{d\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)^2} dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  -1/2*Int[(e*x)^m*E^(-d*(a+b*Log[c*x^n])^2),x] + 1/2*Int[(e*x)^m*E^(d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(-d*(a+b*Log[c*x^n])^2),x] + 1/2*Int[(e*x)^m*E^(d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```